
Summary 
We coded the bond strategy from report #03 for Amibroker and took the following notes: 

• Implementation for Amibroker was straightforward and did not pose any major challenges. 

• Amibroker does not offer the same data sources that TuringTrader does. In particular, we are 

missing access to economic data from FRED. To remedy this, we substituted these time series 

with data available from Norgate. These series are similar in nature, but not identical. 

Consequently, the strategy’s result are not 100% identical to the results we obtained on the 

TuringTrader platform. 

• Amibroker does not offer a convenient method of splicing asset quotes with backfills. 

Consequently, we are not able to backtest the strategy prior to the inception dates of the ETFs 

used. Because we rely on prior research conducted on the TuringTrader platform, this has been 

deemed acceptable. 

• Because the results of our Amibroker implementation inevitably differ from our results on 

TuringTrader, we decided to adjust it’s parameters by brute-force optimization. This steps leads 

to further deviations from the results presented in report #03. 

• Amibroker’s backtester is not a good fit for portfolio strategies. The APIs are optimized for 

simple and speedy execution, but not well suited for complex strategies that rotate through a 

universe of assets and dynamically adjust position sizes. While all of this is possible using a 

custom backtesting procedure, it results in convoluted code, which is hard to read and hard to 

debug. Further, the platform does not offer any features for creating portfolios of strategies. 

• We learned a lot about Amibroker in the past few days, but we are certainly not Amibroker pros. 

Some of the hurdles we have seen might be due to our lack of experience with the platform. 

Also, our implementation might not be as tight as it could be, if developed by a seasoned 

Amibroker developer. Nonetheless, our assessment that the platform is ultimately not well-

suited for complex portfolio strategies and meta-portfolios stands. 

Installation 
To make sure that the strategy works the same way it does on our system, follow these steps: 

Create Watchlist 
The strategy works on a pre-defined universe. The strategy code specifically expects certain ETFs to be 

present. If any of these ETFs are missing, or substituted with other ETFs, the strategy might behave 

differently. 

 

Right click under Watch Lists and choose the option New watchlist. 

 

When prompted for the name of the new watch list enter BeSol-Bond-Rotation. 

Find your newly created watchlist, right click, and select Import. 

 

When prompted to select the watch list, you should find it pre-selected. Click OK to confirm. 

Navigate to the file BeSol-Bond-Rotation.tls that we provided and click Open. 

 

You should see the following eight symbols in you watchlist: JNK, LQD, TLT, TLH, IEF, IEI, BIL, TIP. 

Navigate to the file BeSol-Bond-Rotation.apx that we provided and click Open. 

 

Confirm that the strategy is applied to a Filter, and that the start date is 01/01/2007. 

Click the green filter icon, and confirm that the Filter uses the BeSol-Bond-Rotation watchlist. 

Click the tool icon, and confirm the following settings: 

Most of the settings are defined inside the strategy’s code. However, the following settings must be 

adjusted here: 

• Positions 

• Periodicity 

• Pad and align 

You are now ready to run the strategy. 

Backtest 
In our testing, we achieved the following results with weekly rebalancing: 

Strategy Code 
//============================================================================== 

// Project:     Travis Cook 2022 

// Name:        BeSol-Bond-Rotation.afl 

// Description: Bond Rotation, based on Heine's Bond Trading Model 

// History:     2022iii10, FUB, created 

//------------------------------------------------------------------------------ 

// Copyright:   (c) 2011-2022, Bertram Solutions LLC 

//              http://www.bertram.solutions 

//============================================================================== 

 

//------------------------------------------------------------------------------ 

// strategy parameters 

 

//WKS = 1; // bars per week: 1 for weekly, 5 for daily bars 

switch (Interval(0)) 

{ 

 case inDaily: WKS = 5; break; 

 case inWeekly: WKS = 1; break; 

 default:  

  WKS = 1; // dummy required to display error message 

  Error("must use daily or weekly Interval"); 

  break; 

} 

 

BOND_NUM = 2; //Optimize("BOND_NUM", 2, 1, 3, 1); // default: 2 

BOND_PER = 10 * WKS; //Optimize("BOND_PER", 10*5, 6*WKS, 26*WKS, WKS); // default: 10 weeks 

YIELD_LT_PER = 7 * WKS; //Optimize("YIELD_LT_PER", 7*WKS, 4*WKS, 13*WKS, WKS); // default: 7 

weeks 

YIELD_ST_PER = 10 * WKS; //Optimize("YIELD_ST_PER", 6*5, 4*WKS, 13*WKS, WKS); // default: 8 weeks 

UTILITY_IDX_PER = 4 * WKS; //Optimize("UTILITY_INDEX_PER", 10 * WKS, 3 * WKS, 26 * WKS, WKS); // 

default: 6 weeks 

COMMODITY_IDX_PER = 6 * WKS; //Optimize("COMMODITY_INDEX_PER", 20 * WKS, 3 * WKS, 26 * WKS, WKS); 

// default: 17 weeks 

SAFE_PER = 38 * WKS; //Optimize("SAFE_PER", 38 * WKS, 8 * WKS, 52 * WKS, WKS); // default: 38 

weeks 

SPX_FAST = 2 * WKS; // default: 2 weeks 

SPX_SLOW = 10 * WKS; // default: 10 weeks 

 

//------------------------------------------------------------------------------ 

// strategy universe 

// this strategy must be applied to a watchlist with the following tickers:  

// - JNK: SPDR Bloomberg High Yield Bond ETF (since Dec 2007) 

// - LQD: iShares iBoxx $ Investment Grade Corporate Bond ETF (since Jul 2002) 

// - TLT: iShares 20 Plus Year Treasury Bond ETF (since Jul 2002)  

// - TLH: iShares 10-20 Year Treasury Bond ETF (since Jan 2007) 

// - IEF: iShares 7-10 Year Treasury Bond ETF (since Jul 2002) 

// - IEI: iShares 3-7 Year Treasury Bond ETF (since Jan 2007) 

// - BIL: SPDR Bloomberg 1-3 Month T-Bill ETF (since May 2007) 

// - TIP: iShares TIPS Bond ETF (since Dec 2003) 

 

CHECK_STOCK_MARKET = IIf(StrToUpper(Name()) == "JNK", True, False); 

IS_SAFE_ASSET = IIf(StrToUpper(Name()) == "TIP" || StrToUpper(Name()) == "BIL", True, False); 

 

switch (StrToUpper(Name())) 

{ 

 case "JNK": RAW_SCORE = 7; break; 

 case "LQD": RAW_SCORE = 6; break; 

 case "TLT": RAW_SCORE = 5; break; 

 case "TLH": RAW_SCORE = 4; break; 

 case "IEF": RAW_SCORE = 3; break; 

 case "IEI": RAW_SCORE = 2; break; 

 case "BIL": RAW_SCORE = 1; break; 

 case "TIP": RAW_SCORE = 1; break; 

 default: RAW_SCORE = 0; break; 



#04 Coding 
 

10 
 

} 

 

//------------------------------------------------------------------------------ 

// backtester settings 

// this strategy requires the following settings: 

// - Positions: Long 

// - Periodicy: Weekly 

// - Pad and align all data to reference symbol: $SPX 

 

SetOption("InitialEquity", 1e6); 

SetOption("MarginRequirement", 100); // 100% == pay in full 

 

SetOption("CommissionMode", 3); // mode 3 == $/share 

SetOption("CommissionAmount", 0.01); // $0.01 

 

SetTradeDelays(1, 1, 1, 1); // next day open 

BuyPrice = SellPrice = ShortPrice = CoverPrice = Open; 

SetOption("AllowSameBarExit", False); 

SetOption("ActivateStopsImmediately", False);  

SetOption("UsePrevBarEquityForPosSizing", True); 

SetOption("AllowPositionShrinking" , True);   

 

SetOption("MaxOpenPositions", BOND_NUM); 

// FIXME: leaves 50% cash, if no bond (other than BIL or TIP) is investable 

SetPositionSize( IIf(IS_SAFE_ASSET, 100, 100 / BOND_NUM), spsPercentOfEquity); 

SetOption("MinPosValue", 0.01);  

RoundLotSize = 1; 

 

SetBacktestMode(backtestRotational); 

SetOption("WorstRankHeld", 1); 

 

SetOption("InterestRate", 0.0); 

//SetOption("PortfolioReportMode", 1); // mode 1 == detailed log, mode 2 == summary 

 

//------------------------------------------------------------------------------ 

// manage safe assets 

 

if (IS_SAFE_ASSET) 

{ 

 mom = LinRegSlope(log(Close), SAFE_PER); 

  

 if (StrToUpper(Name()) == "TIP") safe2 = "BIL"; 

 else safe2 = "TIP"; 

  

 SetForeign(safe2); 

  mom2 = LinRegSlope(log(Close), SAFE_PER); 

 RestorePriceArrays(); 

  

 PositionScore = IIf(mom > mom2, RAW_SCORE, 0); 

} 

 

//------------------------------------------------------------------------------ 

// manage investment assets 

 

else 

{ 

 bondPriceRising = IIf(LinRegSlope(log(Close), BOND_PER) > 0, 1, 0); 

  

 SetForeign("%TYX"); // long-term yield (fred:DGS20) 

  ltYieldFalling = IIf(LinRegSlope(Close, YIELD_LT_PER) < 0, 1, 0); 

 RestorePriceArrays(); 

 

 SetForeign("%IRX"); // short-term yield (fred:DTB3) 

  stYieldFalling = IIf(LinRegSlope(Close, YIELD_ST_PER) < 0, 1, 0); 

 RestorePriceArrays(); 

 SetForeign("$DJU"); // Dow Jones Utility Index 

  utilityIdxRising = IIf(LinRegSlope(log(Close), UTILITY_IDX_PER) > 0, 1, 0); 

 RestorePriceArrays(); 

 

 SetForeign("#PPISA"); // Producer Price Index (fred:PPIACO) or CRB Index 

  commodityIdxFalling = IIf(LinRegSlope(log(Close), COMMODITY_IDX_PER) < 0, 1, 0); 

 RestorePriceArrays(); 

 

 if (CHECK_STOCK_MARKET) 

 { 

  SetForeign("$SPXTR"); // S&P 500 Total Return Index 

   stockMarketHealthy = Iif(EMA(Close, SPX_FAST) > EMA(Close, SPX_SLOW), 0, -

100); 

  RestorePriceArrays(); 

 } 

 else 

 { 

  stockMarketHealthy = 0; 

 } 

  

 totalScore = bondPriceRising + stockMarketHealthy 

  + ltYieldFalling + stYieldFalling 

  + utilityIdxRising + commodityIdxFalling; 

   

 PositionScore = IIf(totalScore >= 3, RAW_SCORE, 0); 

} 

 

//============================================================================== 

// end of file 

 


